Newtonian Orbit for Schwarzschild Metric

The purpose of this worksheet is to derive the classical test of General Relativity regarding precession of orbits. We begin with the "energy conservation" equation.

```
> restart: alias(r=r(tau),u=u(phi),up=up(phi)):
```

Recall that the effectve potential can be written in scaled coordinates as

```
> Veff:=-1/r+alpha^2/2/r^2-alpha^2/r^3;
```

where r is in unitsof M and $\alpha = \frac{l}{M}$.

$$Veff := -\frac{1}{r} + \frac{\alpha^2}{2\,r^2} - \frac{\alpha^2}{r^3}$$

The analog to the conservation of energy is given by

```
> Energy:=EE=1/2*diff(r,tau)^2+Veff;
```

$$Energy := EE = \frac{1}{2}\left(\frac{\partial}{\partial\tau} r\right)^2 - \frac{1}{r} + \frac{\alpha^2}{2\,r^2} - \frac{\alpha^2}{r^3}$$

One can differentiate this quantity with respect to the proper time and divide out $\frac{\partial}{\partial\tau} r$ to get a second order equation:

```
> eqn1:=expand(rhs(diff(Energy,tau))/diff(r,tau))=0;
  rtt:=solve(eqn1,diff(r,tau$2)):
```

$$eqn1 := \left(\frac{\partial^2}{\partial\tau^2} r\right) + \frac{1}{r^2} - \frac{\alpha^2}{r^3} + \frac{3\,\alpha^2}{r^4} = 0$$

We will develop a parallel analysis to the Kepler problem by obtaining the orbit equation in terms of $\mathrm{u}(\phi) = \frac{1}{r}$

```
> upp:=-r^2*diff(r,tau$2)/alpha^2:
  eqn2:=subs(diff(r,tau$2) =
  (-r^2+alpha^2*r-3*alpha^2)/r^4,diff(u,phi$2)=upp);
```

$$eqn2 := \frac{\partial^2}{\partial\phi^2} u = -\frac{-r^2 + \alpha^2\,r - 3\,\alpha^2}{r^2\,\alpha^2}$$

The orbit equation takes the form

```
> eqn3:=expand(subs(r=1/u,eqn2));
```

$$eqn3 := \frac{\partial^2}{\partial\phi^2} u = \frac{1}{\alpha^2} - u + 3\,u^2$$

We need to insert the non-geometrized units if we wish to look at small corrections due to the relativistic terms for planetary motion in the solar system.
First, we reintroduce the M scaling of $r = 1/u$.

```
> eqn4:=expand(subs(u=u*M,eqn3)/M);
```

$$eqn4 := \frac{\partial^2}{\partial \phi^2} u = \frac{1}{M \alpha^2} - u + 3 M u^2$$

Now, we recall that alpha is $\frac{l}{M}$.

```
> eqn5:=subs(alpha=l/M,eqn4);
```

$$eqn5 := \frac{\partial^2}{\partial \phi^2} u = \frac{M}{l^2} - u + 3 M u^2$$

M is in geometrized units, so we need to put back the $\frac{G}{c^2}$.

```
> eqn6:=subs(M=G*M/c^2,eqn5);
```

$$eqn6 := \frac{\partial^2}{\partial \phi^2} u = \frac{G M}{c^2 l^2} - u + \frac{3 G M u^2}{c^2}$$

Also, we recall that the angular momentum is rescaled with both *c* and the particle mass. See the classical formalism! This gives the correct dimensional orbit equation with the relativistic correction. compare this to the classical orbit problem.

```
> eqn7:=subs(l=L/m/c,eqn6);
```

$$eqn7 := \frac{\partial^2}{\partial \phi^2} u = \frac{G M m^2}{L^2} - u + \frac{3 G M u^2}{c^2}$$

Note that for there is a correction in the last terms as $\frac{GM}{c^2}$ is small. The first term on the right hand side is the classical constant term. We can write this in terms of these parameters as

```
> EQ:=diff(u,phi$2)+u = lambda+delta*u^2;
```

$$EQ := \left(\frac{\partial^2}{\partial \phi^2} u\right) + u = \lambda + \delta u^2$$

There is no exact solution, but since δ is small, we can obtain anapproximate solution using a perturbation analysis. First we set $\delta = 0$ and find theexact solution.

```
> EQ0:=subs(delta=0,EQ);
```

$$EQ0 := \left(\frac{\partial^2}{\partial \phi^2} u\right) + u = \lambda$$

We call *u1* the solution to this equation. To get this solution of the Kepler problem we use simple ODEs. u = solution to the homogeneous equation in terms of sines and cosines plus $u = \lambda$:

```
> dsolve(EQ0);
```

$$u = \sin(\phi)\ \_C2 + \cos(\phi)\ \_C1 + \lambda$$

Setting the initial conditions appropriately, we have that $u = \lambda\,(1 + \varepsilon \cos(\phi))$. We verfiy this solution by inserting it into the left hand side of the equation. Yup! It works.

```
> u1:=lambda*(1+epsilon*cos(phi));
  lhs(EQ0)=simplify(subs(u=u1,lhs(EQ0)));
```

$$u1 := \lambda\,(1 + \varepsilon \cos(\phi))$$

$$\left(\frac{\partial^2}{\partial\phi^2}u\right)+u=\lambda$$

Before we continue, we plot asolution. The solution is actually a general form for a conic in polar coordinates.

```
> R:=1/u1:
  lambda:=2:
  epsilon:=0.7:
  plot([R,phi,phi=0..2*Pi],coords=polar);
```



Now we let the solution to the main problem be this exact solution plus a correction good to order δ^2. Let $u = u1 + up$. Inserting this into equation EQ gives an equation for the correction *up*

```
> lambda:='lambda':
  epsilon:='epsilon':
  EQ1:=simplify(subs(u=u1+up,EQ));
```

$$EQ1 := \left(\frac{\partial^2}{\partial\phi^2}up\right)+\lambda+up = \lambda+\delta\lambda^2+2\,\delta\lambda^2\varepsilon\cos(\phi)+2\,\delta\lambda\,up+\delta\lambda^2\varepsilon^2\cos(\phi)^2+2\,\delta\lambda\varepsilon\cos(\phi)\,up+\delta\,up^2$$

Skipping any details, we *miraculously* dream up the approximate solution

```
> up:=delta*lambda^2*((1+epsilon^2/2)+epsilon*phi*sin(phi)-epsilon^2
  /6*cos(2*phi));
```

$$up := \delta\lambda^2\left(1+\frac{\varepsilon^2}{2}+\varepsilon\phi\sin(\phi)-\frac{1}{6}\varepsilon^2\cos(2\,\phi)\right)$$

Inserting this expression into the equation EQ1, we find that the left and right hand sides differ by terms of order δ^2:

```
> factor(lhs(expand(EQ1))-rhs(expand(EQ1)));
```

$$-\frac{1}{9}\lambda^3\left(3+2\,\varepsilon^2-\varepsilon^2\cos(\phi)^2+3\,\varepsilon\phi\sin(\phi)\right)\delta^2\left(3\,\lambda\varepsilon\sin(\phi)\,\delta\phi+2\,\delta\varepsilon^2\lambda-\delta\varepsilon^2\lambda\cos(\phi)^2+3\,\delta\lambda+6\,\varepsilon\cos(\phi)+6\right)$$

Therefore, we have the approximate solution given next.

```
> u:=u1+up;
```

$$u := \lambda\left(1+\varepsilon\cos(\phi)\right)+\delta\lambda^2\left(1+\frac{\varepsilon^2}{2}+\varepsilon\phi\sin(\phi)-\frac{1}{6}\varepsilon^2\cos(2\,\phi)\right)$$

Actually, standard perturbation analysis, which you are generally not exposed to, tells us that this solution agrees with the following function to order δ^2:

```
> U:=lambda*(1+epsilon*(cos(phi-delta*lambda*phi)));
```

$$U := \lambda\,(1 + \varepsilon\cos(-\phi + \delta\,\lambda\,\phi))$$

This can be confirmed by doing a series expasion about $\delta = 0$:

```
> series(U,delta=0,2);
```

$$\lambda\,(1 + \varepsilon\cos(\phi)) + \varepsilon\sin(\phi)\,\lambda^2\,\phi\,\delta + O(\delta^2)$$

It is now time to plot the perturbed solution. We repeat the plot from above except we add in the correction. Note that the orbit does not close upon itself in one revolution! It precesses. The deviation from precession is given by $\Delta = 2\,\pi\,\delta\,\lambda$.

```
> R:=1/U:
  lambda:=1.5:
  epsilon:=0.75:
  delta:=0.01:
  plot([R,phi,phi=0..10*Pi],coords=polar,numpoints=200);
```



The deviation is given by (here we need $\delta = \frac{3\,G\,M}{c^2}$, $\lambda = \frac{G\,M\,m^2}{L^2}$, and $L^2 = \frac{M\,m\,k\,a\,(1-e^2)}{M+m}$.)

```
> Delta:=6*Pi*G*M/(a*c^2*(1-e^2));
```

$$\Delta := \frac{6\,\pi\,G\,M}{a\,c^2\,(1-e^2)}$$

Example: Units in MKS

```
> Digits:=10: G:=6.67428*10^(-11): M:=1.9891*10^(30):
  c:=2.99792458*10^8:
```

Solar System Data

```
> e:=0.205630 : a:=57909100000: T:=87.9691/365.25:
```

Δ is the number of radians per orbit of Mercury. About 365/88 revolutions occur in one of Earth's years. Now convert radians to seconds of an arc and multiply by 100 to get deviation over a century.

```
> 'Delta'=evalf(Delta); evalf(rhs(%)/T): evalf(%*180/Pi)*60*60*100;
```

$$\Delta = 0.5020385437\ 10^{-6}$$

$$42.99542734$$

```
>
>
```